

An Application of Gröbner Bases

Shengxiang Xia¹ and Gaoxiang Xia²

Abstract. In this paper, we program a procedure using Maple's packages, with it we can realize mechanical proving of some theorems in elementary geometry.

Keywords: Gröbner base; geometry theorem; Maple.

AMS Subject Classification (2000) 68-04, 68N30

1. Introduction.

Gröbner bases have been fruitfully applied to many problems, one of them is to deal with the problem of automated geometry theorem proving (see [1], [2], [3] and [4]). Maple is a comprehensive computer system for advanced mathematics, and its Gröbner package can be used to compute a Gröbner basis of a polynomial ring. In this paper, we will discuss how to use Maple's packages to realize mechanization of theorem-proving in elementary geometry. Generally, the problem of mechanical theorem-proving can be done by the following three steps.

(1) The first step is to introduce a number system and a coordinate system such that a theorem of elementary geometry can be changed into an algebraic problem;

(2) The second step is to sort the algebraic expressions of the involved theorem's conditions, to set measures for checking if algebraic expressions of the involved theorem's result can be induced from the sorted algebraic expressions;

(3) The last step is to compile a program according to the above measures, and carry out it on a computer.

¹ College of Science, Shandong Jianzhu University, Jinan 250101, China
E-mail: summer_5069@yahoo.com.cn

² Shandong Institute of Architectural Science, Jinan 250031, China

2. Some Algebraic Expressions of Common Geometric Relations.

The method of Gröbner bases depends on the construction of a particular type of polynomials to represent given geometric relations. To illustrate the translation of geometric statements into a suitable system of polynomials, we consider a simple example: the line A_1A_2 is parallel to the line A_3A_4 . Let the coordinates of points A_j be (x_{A_j}, y_{A_j}) , $j=1,2,3,4$, in a coordinate system. In an analytic setting two lines are parallel if and only if they have the same slope. We can translate $A_1A_2 // A_3A_4$ into an equation by relating their slopes,

$$\frac{y_{A_1} - y_{A_2}}{x_{A_1} - x_{A_2}} = \frac{y_{A_3} - y_{A_4}}{x_{A_3} - x_{A_4}}.$$

In the form of a polynomial equation this condition is $(x_{A_1} - x_{A_2})(y_{A_3} - y_{A_4}) - (y_{A_1} - y_{A_2})(x_{A_3} - x_{A_4}) = 0$.

Although it may seem as if each polynomial function needs to be set equal to zero, this is not required for the Gröbner basis method. Hence in order to generalize these Maple functions for use in the method of Gröbner basis each Maple function returns only a polynomial in x_i and y_i . In a Maple command window, we input common geometric relations, and save the functions as a Maple internal file for conveniently using them later.

```

> parallel:=(A1, A2, A3, A4)->(x[A1]-x[A2])*(y[A3]-y[A4])-(y[A1]-y[A2])*(x[A3]-x[A4]);
parallel = (A1, A2, A3, A4) -> (x_{A1} - x_{A2}) (y_{A3} - y_{A4}) - (y_{A1} - y_{A2}) (x_{A3} - x_{A4})

> vertical:=(A1, A2, A3, A4)->(x[A1]-x[A2])*(x[A3]-x[A4])+(y[A1]-y[A2])*(y[A3]-y[A4]);
vertical = (A1, A2, A3, A4) -> (x_{A1} - x_{A2}) (x_{A3} - x_{A4}) + (y_{A1} - y_{A2}) (y_{A3} - y_{A4})

> distance2:=(A1, A2)->(x[A1]-x[A2])^2+(y[A1]-y[A2])^2 ;
distance2 = (A1, A2) -> (x_{A1} - x_{A2})^2 + (y_{A1} - y_{A2})^2

> collinear:=(A1, A2, A3)->(x[A2]-x[A1])*(y[A3]-y[A1])-(x[A3]-x[A1])*(y[A2]-y[A1]);
collinear = (A1, A2, A3) -> (x_{A2} - x_{A1}) (y_{A3} - y_{A1}) - (x_{A3} - x_{A1}) (y_{A2} - y_{A1})

> midpoint:=(A1, A2, A3)->(2*x[A3]-x[A1]-x[A2], 2*y[A3]-y[A1]-y[A2]);
midpoint = (A1, A2, A3) -> 2 x_{A3} - x_{A1} - x_{A2}, 2 y_{A3} - y_{A1} - y_{A2}

> proportion:=(A1, A2, A3, r)->((1+r)*x[A3]-x[A1]-r*x[A2], (1+r)*y[A3]-y[A1]-r*y[A2]);
proportion = (A1, A2, A3, r) -> (1 + r) x_{A3} - x_{A1} - r x_{A2}, (1 + r) y_{A3} - y_{A1} - r y_{A2}

> isogonal:=(A1, A2, A3, A4, A5, A6)->((y[A3]-y[A2])*(x[A1]-x[A2])-(y[A1]-y[A2])*(x[A3]-x[A2]))*((x[A6]-x[A5])*
(x[A4]-x[A5])+(y[A6]-y[A5])*(y[A4]-y[A5]))-((y[A6]-y[A5])*(x[A4]-x[A5])-(y[A4]-y[A5])*(x[A6]-x[A5]))*
((x[A3]-x[A2])*(x[A1]-x[A2])+(y[A3]-y[A2])*(y[A1]-y[A2]));
isogonal = (A1, A2, A3, A4, A5, A6) -> ((y_{A3} - y_{A2}) (x_{A1} - x_{A2}) - (y_{A1} - y_{A2}) (x_{A3} - x_{A2})) ((x_{A6} - x_{A5}) (x_{A4} - x_{A5}) + (y_{A6} - y_{A5}) (y_{A4} - y_{A5}))
- ((y_{A6} - y_{A5}) (x_{A4} - x_{A5}) - (y_{A4} - y_{A5}) (x_{A6} - x_{A5})) ((x_{A3} - x_{A2}) (x_{A1} - x_{A2}) + (y_{A3} - y_{A2}) (y_{A1} - y_{A2}))

> onbisector:=(A1, A2, A3, A4)->((y[A4]-y[A2])*(x[A1]-x[A2])-(y[A1]-y[A2])*(x[A4]-x[A2]))*((x[A3]-x[A2])*
(x[A4]-x[A2])+(y[A3]-y[A2])*(y[A4]-y[A2]))-((y[A3]-y[A2])*(x[A4]-x[A2])-(y[A4]-y[A2])*(x[A3]-x[A2]))
*((x[A4]-x[A2])*(x[A1]-x[A2])+(y[A4]-y[A2])*(y[A1]-y[A2]));
onbisector = (A1, A2, A3, A4) -> ((y_{A4} - y_{A2}) (x_{A1} - x_{A2}) - (y_{A1} - y_{A2}) (x_{A4} - x_{A2})) ((x_{A3} - x_{A2}) (x_{A4} - x_{A2}) + (y_{A3} - y_{A2}) (y_{A4} - y_{A2}))
- ((y_{A3} - y_{A2}) (x_{A4} - x_{A2}) - (y_{A4} - y_{A2}) (x_{A3} - x_{A2})) ((x_{A4} - x_{A2}) (x_{A1} - x_{A2}) + (y_{A4} - y_{A2}) (y_{A1} - y_{A2}))

> save parallel, vertical, distance2, collinear, midpoint, proportion, isogonal, onbisector, "C:\\geometry.m";

```

3. Definitions and a Basic Principle.

The method in this paper is based on the theory of Gröbner bases, so we introduce some concepts and a theorem about Gröbner bases.

Definition 1. Let I be a nonzero ideal of a ring A , $G = \{g_1, \dots, g_s\}$ be a nonzero finite set of polynomials. The G is called a Gröbner basis of the ideal I , if and only if for each polynomial f in I , there exists j , $1 \leq j \leq s$, such that $lp(g_j) | lp(f)$. Where $lp(f)$ denotes the leader product of power of f .

Definition 2. For polynomials f, g, h in a ring A , $g \neq 0$, f is called one-step reduce to h by module g , denoted by $f \xrightarrow{g} h$, if and only if $lp(g)$ is a factor of nonzero monomial expression X of f , and $h = f - \frac{X}{lt(g)}g$, where $lt(g)$ denotes the leader of g .

Definition 3. Let f, f_1, \dots, f_s , and $f_j \neq 0$ ($1 \leq j \leq s$), set $F = \{f_1, \dots, f_s\}$, f is called reduce to h about module F , denoted by $f \xrightarrow{F} h$, if and only if $f \xrightarrow{f_1} h_1 \xrightarrow{f_2} h_2 \rightarrow \dots \xrightarrow{f_t} h_t = h$, where $f_{ij} \in F, h_{ij} \in A$ ($j=1, \dots, t$).

Theorem A. (Buchberger's Theorem [5]) Let I be a nonzero ideal of a ring $A = [x_1, \dots, x_n]$, $G = \{f_1, \dots, f_t\} \subseteq I \setminus \{0\}$, then the followings are equivalent:

- (a) G is a Gröbner basis of I ;
- (b) $f \in I$ if and only if $f \xrightarrow{G} 0$.

4. The Main Procedure.

We have seen that we can translate conditions and the conclusion of a geometric theorem into polynomials: f_1, \dots, f_m (the hypotheses) and g (the conclusion) in the ring $K[x_1, \dots, x_j; y_1, \dots, y_j]$. In what sense then does our conclusion g , follow from the hypotheses f_1, \dots, f_m ? An algebraic formulation of the problem is as follows:

$$\forall (x_1, \dots, x_j, y_1, \dots, y_j), (f_1 = 0 \wedge \dots \wedge f_m = 0) \Rightarrow g = 0.$$

Let I be the ideal generated by the set $\{f_1, \dots, f_m\}$ in $K[x_1, \dots, x_j; y_1, \dots, y_j]$, then the conclusion g follows from the hypotheses f_1, \dots, f_m means that $g \in I$.

Using the command **gbasis** of Maple, we can calculate a Gröbner basis of the ideal I , and a Gröbner basis of an ideal has the property that every polynomial in the ideal reduces to 0 with respect to the basis. Hence, to determine if the conclusion g is in the ideal I , we need only to use the command **normalf** of Maple to calculate the remainder of the conclusion polynomial g after division by a Gröbner basis of the ideal I .

By above algorithmic principle, we compile a procedure **provegeo** in Maple language, with it we can realize mechanization of some geometry theorems proving. The source codes of the procedure 'provegeo' are shown in Maple sheets of following examples. To use the procedure conveniently, we may save it as a Maple package.

5. Examples

Example 1. The diagonals of a rhombus are mutual perpendicular.

Establishing a coordinate system as the figure 1, then we have the following conditions:

1. $AD \parallel BC$,
2. $AD = AB$,
3. The parallelogram $ABCD$ is non-degenerate.

Our conclusion is $AC \perp BD$. The mechanical

proof is as following Maple sheet:




Figure 1

```

C:\Program Files\Maple 9\examples\example1-check maple sheet.mw
File Edit View Insert Format Tools Window Help
[Icons]
> restart;
> read "c:\geometry.m";
> x[A]:=0: y[A]:=0: y[B]:=0:
> y[D]:=y[C]:
> c1:=parallel(A, D, B, C);
                                     c1 := xD yC + yC (xB - xC)
> c2:=distance2(A, D)-distance2(A, B);
                                     c2 := xD2 + yC2 - xB2
> c3:=x[B]*y[C]-a;
                                     c3 := xB yC - a
> r:=vertical(A, C, B, D);
                                     r := -xC (xB - xD) + yC2
> c:=[c1, c2, c3];
                                     c := [ xD yC + yC (xB - xC), xD2 + yC2 - xB2, xB yC - a ]
> vars:=[x[B], x[C], y[C]];
                                     vars := [xB, xC, yC]
> with(Groebner):
> provegeo:=proc(conditions, conclusion, x)
  local T, gb, rem;
  T:=op(1..nops(x), x);
  gb:=gbasis(conditions, tdeg(T));
  rem:=normalf(conclusion, gb, tdeg(T));
  if rem=0 then
    print (true);
  else
    print (rem);
  fi;
end:
> provegeo(c, r, vars);
                                     true

```

Remark: In the above proof, the non-generate condition $x_B y_C \neq 0$ (i.e. $x_B y_C \neq 0$) of the parallelogram $ABCD$ should be looked as a given condition, otherwise the proposition could not be checked correctly. The variable x_D is not an independent variable, it is as a parameter. If x_D is listed in 'vars', the proposition is also checked to be true.

Example 2 (Apollonius' theorem). Given a $\triangle ABC$, if D is any point on BC such that it divides BC in the ratio $n:m$ ($mBD = nDC$), then $mAB^2 + nAC^2 = mBD^2 + nDC^2 + (m+n)AD^2$. When $m=n(=1)$, that is, AD is the median falling on BC , the theorem reduces to $AB^2 + AC^2 = BD^2 + DC^2 + 2AD^2 = \frac{1}{2}BC^2 + 2AD^2$.

Establishing a coordinate system as the figure 2, then we have the following conditions:

1. D divides BC in the ratio $n : m$,
2. $\triangle ABC$ is non-generate.

Our conclusion is $mAB^2 + nAC^2 = mBD^2 + nDC^2 + (m+n)AD^2$.

The mechanical proof is as following Maple sheet:




Figure 2

```

C:\Program Files\Maple 9\examples\example2-check maple sheet.mw
File Edit View Insert Format Tools Window Help
[Icons]
> restart;
> read 'c:\geometry.m';
> x[A]:=0:y[A]:=0:y[B]:=0:
> c1:=proportion(B,C,D,n/m);
c1 = (1 + n/m) x_D - x_B - n x_C / m, (1 + n/m) y_D - n y_C / m
> c2:=x[B]*y[C]-a;
c2 = x_B y_C - a
> res:=m*distance2(A,B)+n*distance2(A,C)-m*distance2(B,D)-n*(distance2(D,C))- (m+n)*distance2(A,D);
res = m x_B^2 + n (x_C^2 + y_C^2) - m ((x_B - x_D)^2 + y_D^2) - n ((x_D - x_C)^2 + (y_D - y_C)^2) - (m + n) (x_D^2 + y_D^2)
> vars:=[x[B], x[C], y[C], x[D], y[D]];
vars = [x_B, x_C, y_C, x_D, y_D]
> con:=[c1, c2];
con = [(1 + n/m) x_D - x_B - n x_C / m, (1 + n/m) y_D - n y_C / m, x_B y_C - a]
> with(Groebner):
> provegeo:=proc(conditions, conclusion, x)
local T, gb, rem;
T:=op(1..nops(x), x);
gb:=gbasis(conditions, tdeg(T));
rem:=normalf(conclusion, gb, tdeg(T));
if rem=0 then
print(true);
else
print(rem);
fi;
end;
> provegeo(con, res, vars);
true

```

Example 3. Supposed that CD bisects $\angle ACB$, and $AE \parallel CD$, then $\triangle ACE$ is an isosceles triangle.

Establishing a coordinate system as the figure 3,

then we have the following conditions:

1. CD bisects $\angle ACB$,
2. A, B, D are collinear,
3. $AE \parallel CD$,
4. $\triangle ABC$ is non-generate.




Figure 3

Our conclusion is $AC = CE$. The mechanical proof is as following Maple sheet:

```

C:\Program Files\Maple 9\examples\example3-check maple sheet.mw
File Edit View Insert Format Tools Window Help
[Icons]

> restart;
> read "c:\\geometry.m";
> x[B]:=0: y[B]:=0: y[C]:=0: y[E]:=0:
> c1:=onbisector(A,C,B,D);
c1 = -(yD(xA-xC)-yA(xD-xC))xC(xD-xC)-yDxC((xD-xC)(xA-xC)+yDyA)

> c2:=parallel(A,E,C,D);
c2 = -(xA-xB)yD-yA(xC-xD)

> c3:=collinear(A,B,D);
c3 = -xA(yD-yA)+(xD-xA)yA

> c4:=y[A]*x[C]-a;
c4 = yAxC-a

> vars:=[x[A],y[A],x[C],x[D],y[D],x[E]];
vars = [xA,yA,xC,xD,yD,xB]

> r:=distance2(C,E)-distance2(A,C);
r = (xC-xB)2 - (xA-xC)2 - yA2

> c:=[c1,c2,c3,c4];
c = [-(yD(xA-xC)-yA(xD-xC))xC(xD-xC)-yDxC((xD-xC)(xA-xC)+yDyA), -(xA-xB)yD-yA(xC-xD), -xA(yD-yA)+(xD-xA)yA, yAxC-a]

> with(Groebner):
> provegeo:=proc(conditions, conclusion, x)
local T, gb, rem;
T:=op(1..nops(x), x);
gb:=gbasis(conditions, tdeg(T));
rem:=normalf(conclusion, gb, tdeg(T));
if rem=0 then
print (true);
else
print (rem);
fi;
end:
> provegeo(c, r, vars);
true

```

Example 4 (Simson line). Given any $\triangle ABC$ and a point P in the plane of the triangle, if perpendiculars from P on to the sides AB, AC, BC , meet those sides at U, V, W respectively, then U, V, W are collinear if and only if P lies on the circumcircle of $\triangle ABC$.

Establishing a coordinate system as the figure 4, for the sufficiency, we have the following conditions:

1. P, B, C lie on the circumcircle of $\triangle ABC$,
2. $PU \perp AB, PV \perp AC, PW \perp BC$,
3. V lies on the line AC, W lies on the line BC ,
4. $\triangle ABC$ is non-generate.

Our conclusion is that U, V, W are collinear. The mechanical proof is as following Maple sheet:




Figure 4

```

C:\Program Files\Maple 9\examples\example4-check maple sheet.mw
File Edit View Insert Format Tools Window Help
[Icons]
> restart;
> read "c:\geometry.m";
> x[A]:=0: y[A]:=0: y[B]:=0: y[U]:=0:
> x[0]:=a: y[0]:=b:
> c1:=distance2(P,0)-distance2(A,0);
c1 := (x_P - a)^2 + (y_P - b)^2 - a^2 - b^2
> c2:=distance2(B,0)-distance2(A,0);
c2 := (x_B - a)^2 - a^2
> c3:=distance2(C,0)-distance2(A,0);
c3 := (x_C - a)^2 + (y_C - b)^2 - a^2 - b^2
> c4:=vertical(P,U,A,B);
c4 := -(x_P - x_U) x_B
> c5:=vertical(P,V,A,C);
c5 := -(x_P - x_V) x_C - (y_P - y_V) y_C
> c6:=vertical(P,W,C,B);
c6 := (x_P - x_W) (x_C - x_B) + (y_P - y_W) y_C
> c7:=collinear(A,C,V);
c7 := x_C y_V - x_V y_C
> c8:=collinear(B,C,W); c9:=x[B]*y[C]-d; ;
c8 := (x_C - x_B) y_W - (x_W - x_B) y_C
c9 := x_B y_C - d

```

```

> con:=[c1, c2, c3, c4, c5, c6, c7, c8, c9]:
> vars:=[x[B], x[C], y[C], x[U], x[V], y[V], x[W], y[W], x[P]];
                                                    vars := [x_B, x_C, y_C, x_U, x_V, y_V, x_W, y_W, x_P]
> with(Groebner):
> provegeo:=proc(conditions, conclusion, x)
  local T, gb, rem;
  T:=op(1..nops(x), x);
  gb:=gbasis(conditions, tdeg(T));
  rem:=normalf(conclusion, gb, tdeg(T));
  if rem=0 then
  print (true);
  else
  print (rem);
  fi;
  end:
> result:=collinear(U, V, W);
                                                    result := (x_V - x_U)y_W - (x_W - x_U)y_V
> provegeo(con, result, vars);
                                                    true

```

In fact, since P lies on the circumcircle of $\triangle ABC$, y_P is not an independent variable, y_P is not listed in 'vars', in this case, the computation time is about 27 seconds. If y_P is listed in 'vars', the proposition is correctly checked, but the computation time is about 103 seconds.


For the necessity, one condition is that U, V, W are collinear, and the conclusion is that P lies on the circumcircle of $\triangle ABC$. Exchanging $c1$ for $result$ in the proof of the sufficiency, we can similarly check that the necessity is true.

Example 5 (Euler line).

In any triangle ΔABC , the orthocenter H , the centroid G and the circumcenter O are collinear, and $GH=2OG$. The line passing by these points is known as the Euler line of ΔABC .

Establishing a coordinate system as the figure 5, we have the following conditions:

1. O is the circumcenter of ΔABC ,
2. G is the centroid of ΔABC ,
3. H is the orthocenter of ΔABC ,
4. ΔABC is non-generate.



Our conclusion is that O, G, H are collinear and $GH = 2OG$. The mechanical proof is as following Maple sheet:

```

> c11:=collinear(A,H,U);
c11 = -(xH-xA)yA - (xU-xA)(yH-yA)

> c12:=collinear(B,H,V);
c12 = xHyV - xVyH

> c13:=collinear(C,H,W);
c13 = (xH-xC)yW - (xW-xC)yH

> c14:=x[C]*y[A]-a;
c14 = xCyA - a

> vars:=[x[A],y[A],x[C],x[U],x[V],y[V],x[W],y[W],x[O],y[O],x[G],y[G],x[H],y[H]];
vars = [xA,yA,xC,xU,xV,yV,xW,yW,xO,yO,xG,yG,xH,yH]

> res1:=collinear(G,H,O);
res1 = (xH-xG)(yO-yG) - (xO-xG)(yH-yG)

> con:=[c1, c2, c3, c4, c5, c6, c7, c8, c9, c10, c11, c12, c13, c14]:
> with(Groebner):
> provegeo:=proc(conditions, conclusion, x)
local T, gb, rem;
T:=op(1..nops(x), x);
gb:=gbasis(conditions, tdeg(T));
rem:=normalf(conclusion, gb, tdeg(T));
if rem=0 then
print (true);
else
print (rem);
fi;
end:
> provegeo(con, res1, vars);
true

> res2:=distance2(G,H)-4*distance2(O,G);
res2 = (xG-xH)2 + (yH+yG)2 - 4(xO-xG)2 - 4(yO-yG)2

> provegeo(con, res2, vars);
true

```

Reporter Time: 3.00s

```

> c10:=collinear(A,B,W);
c10 = -xA(yW-yA) + (xW-xA)yA

```

The above five examples have been checked correctly on a microcomputer, in a similar way, other geometric propositions may be proved by the Maple internal file 'geometry.m' and the procedure

'provegeo'. An appropriate coordinate system should be chosen to reduce variables as possible, and only those independent variables are listed in 'vars', the less variables in 'vars', the less time in the computation.

References

- [1] Franz Winkler, Gröbner bases in geometry theorem proving and simplest dedgeneracy conditions, *Mathematica Pannonica*, 1 (1990), 15-32.
- [2] Joran Elias, Automated geometric theorem proving: Wu's method, *The Montana Mathematics Enthusiast*, 3 (2006), 3-50.
- [3] B. Buchberger and F.Winkler, Gröbner bases and applications, Cambridge University Press, Cambridge, 1998, pp. 3-31 (London Mathematical Society Lecture Note Series. 251).
- [4] Conrad T. Miller, Automated theorem proving in plane geomtry, <http://people.unt.edu/ctm0055/Paper2.pdf> (2006).
- [5] Liu Mulan, *The theory of Gröbner basis and its applications*, Science Press, Beijing, 2000.
- [6] Wu Wenjun, *Basic principles on the mechanization of theorem-proving in elementary geometry*, Science Press, Beijing, 1984.

Xia & Xia